Document ID: /fineweb-2-swissfilter-quality_10-filterrobots/filtered/03365.jsonl.gz/2567

Hallo Christoph,
wenn Du Dich vor einem kleinen AppleScript nicht scheust ...
Das folgende AppleScript sucht alle Textrahmen mit Übersatz.
Am Ende spuck es Dir eine Auflistung der Seiten aus, auf welchen sich Textrahmen mit Übersatz befindent. Wenn Du willst, kannst Du die unten angegebene Zeile auskommentieren und du bekommst zusätzlich zur Seitenangabe noch die erste Zeile des betreffenden Textrahmens angezeigt.
Auf die Schnelle ging's am einfachsten mit AppleScript.
Mit JavaScript (für WInDesing) müßte man wahrscheinlich alle Textrahmen abklappern.
--Ueberlaufer.as
tell application "InDesign CS"
set myDoc to document 1
set myPages to ""
tell myDoc
set myOversetFrames to text frames whose overflows is true
if (count of myOversetFrames) > 0 then
repeat with i from 1 to count of myOversetFrames
--Über die folgende Zeile werden nur die Seiten erfaßt, auf welchen sich Textrahmen mit Übersatz finden
set myPages to myPages & return & name of parent of item i of myOversetFrames
--Über die folgende Zeile werden die Seiten erfaßt, auf welchen sich Textrahmen mit Übersatz finden.
--Zusätzlich wird noch der inhalt der ersten Zeile angegeben (evenutell sinnvoll, wenn sich mehrere Textrahmen mit Übersatz auf einer Seite befinden).
--set myPages to myPages & return & name of parent of item i of myOversetFrames & ": " & first line of item i of myOversetFrames
end repeat
--springt auf die Seite mit dem ersten Vorkommen
set active page of layout window 1 to parent of item 1 of myOversetFrames
display dialog "Überläufer auf folgenden Seiten:" & return & myPages
else
display dialog "Keine Überläufer"
end if
end tell
end tell
--Ende
Grüßle
Martin
als Antwort auf: [#136475]
(Dieser Beitrag wurde von MartinF am 14. Jan 2005, 21:13 geändert)